

OutOfWorld Template

B/ooowt

Purpose

To implement some of the out of world actions.

B/ooowt. §1 Perform Undo; §2 Announce Score Rule; §3 Switch Score Notification On Rule; §4 Standard Report Switching Score Notification On Rule; §5 Switch Score Notification Off Rule; §6 Standard Report Switching Score Notification Off Rule; §7 Prefer Sometimes Abbreviated Room Descriptions Rule; §8 Standard Report Prefer Sometimes Abbreviated Room Descriptions Rule; §9 Prefer Unabbreviated Room Descriptions Rule; §10 Standard Report Prefer Unabbreviated Room Descriptions Rule; §11 Prefer Abbreviated Room Descriptions Rule; §12 Standard Report Prefer Abbreviated Room Descriptions Rule; §13 Announce Pronoun Meanings Rule

§1. Perform Undo.

```
[ Perform_Undo;
  #ifdef PREVENT_UNDO; L__M(##Miscellany, 70); return; #endif;
  if (turns == 1) { L__M(##Miscellany, 11); return; }
  if (undo_flag == 0) { L__M(##Miscellany, 6); return; }
  if (undo_flag == 1) { L__M(##Miscellany, 7); return; }
  if (VM_Undo() == 0) L__M(##Miscellany, 7);
];
```

§2. Announce Score Rule.

```
[ ANNOUNCE_SCORE_R;
  if (actor ~= player) rfalse;
  #ifdef NO_SCORING; L__M(##Score, 2);
  #ifnot; GL__M(##Score); PrintRank();
  #endif;
];
```

§3. Switch Score Notification On Rule.

```
[ SWITCH_SCORE_NOTIFY_ON_R;
  if (actor ~= player) rfalse;
  #ifdef NO_SCORING; ANNOUNCE_SCORE_R();
  #ifnot; notify_mode=1; #endif;
];
```

§4. Standard Report Switching Score Notification On Rule.

```
[ REP_SWITCH_NOTIFY_ON_R;
  if (actor ~= player) rfalse;
  #ifdef NO_SCORING; GL__M(##NotifyOn); #endif;
];
```

§5. Switch Score Notification Off Rule.

```
[ SWITCH_SCORE_NOTIFY_OFF_R;
  if (actor ~= player) rfalse;
  #ifdef NO_SCORING; ANNOUNCE_SCORE_R();
  #ifnot; notify_mode=0; #endif;
];
```

§6. Standard Report Switching Score Notification Off Rule.

```
[ REP_SWITCH_NOTIFY_OFF_R;
  if (actor ~= player) rfalse;
  #ifndef NO_SCORING; GL_M(##NotifyOff); #endif;
];
```

§7. Prefer Sometimes Abbreviated Room Descriptions Rule.

```
[ PREFER_SOMETIMES_ABBREVIATED_R;
  if (actor ~= player) rfalse;
  lookmode=1;
]; ! Brief
```

§8. Standard Report Prefer Sometimes Abbreviated Room Descriptions Rule.

```
[ REP_PREFER_SOMETIMES_ABBR_R;
  if (actor ~= player) rfalse;
  print (string) Story; GL_M(##LModel);
]; ! Brief
```

§9. Prefer Unabbreviated Room Descriptions Rule.

```
[ PREFER_UNABBREVIATED_R;
  if (actor ~= player) rfalse;
  lookmode=2;
]; ! Verbose
```

§10. Standard Report Prefer Unabbreviated Room Descriptions Rule.

```
[ REP_PREFER_UNABBREVIATED_R;
  if (actor ~= player) rfalse;
  print (string) Story; GL_M(##LMode2);
]; ! Verbose
```

§11. Prefer Abbreviated Room Descriptions Rule.

```
[ PREFER_ABBREVIATED_R;
  if (actor ~= player) rfalse;
  lookmode=3;
]; ! Superbrief
```

§12. Standard Report Prefer Abbreviated Room Descriptions Rule.

```
[ REP_PREFER_ABBREVIATED_R;
  if (actor ~= player) rfalse;
  print (string) Story; GL__M(##LMode3);
]; ! Superbrief
```

§13. Announce Pronoun Meanings Rule.

```
[ ANNOUNCE_PRONOUN_MEANINGS_R x y c d;
  if (actor ~= player) rfalse;
  GL__M(##Pronouns, 1);
  c = (LanguagePronouns-->0)/3;
  if (player ~= selfobj) c++;
  if (c==0) return GL__M(##Pronouns, 4);
  for (x = 1, d = 0 : x <= LanguagePronouns-->0: x = x+3) {
    print "~", (address) LanguagePronouns-->x, "~ ";
    y = LanguagePronouns-->(x+2);
    if (y == NULL) GL__M(##Pronouns, 3);
    else { GL__M(##Pronouns, 2); print (the) y; }
    d++;
    if (d < c-1) print ", ";
    if (d == c-1) print (string) LISTAND__TX;
  }
  if (player ~= selfobj) {
    print "~", (address) ME1__WD, "~ "; GL__M(##Pronouns, 2);
    c = player; player = selfobj;
    print (the) c; player = c;
  }
  ".";
];
```